Материалы для самопроверки к теме «Алгоритмы и программы»

В помощь учителю предлагаются листинги программ на PascalABC.NET.
Задачи №4-11 из учебника 10 класса.
4. Дано натуральное трехзначное число n, в записи которого нет нулей. По предложенной записи алгоритма (псевдокод и в виде блок-схемы), определите результаты работы алгоритма.

В приведенной записи DIV и MOD соответственно операции деления нацело и получения остатка от целочисленного деления. В фигурных скобках записаны пояснения (комментарии) к операторам.

	
	Read(n) {Ввод N}

A := n mod 10 {разряд единиц}

B := n div 100 {разряд сотен}

C := n div 10 mod 10 {десятки}

L := (n mod A=0) and (n mod B=0) and (n mod C=0)

Write ( L) {Вывод L}




Переведите алгоритм на язык программирования и получите работающую программу.
program ch4_4;

var 

   n,a,b,c: word;{беззнаковое целое}
   l : boolean;

begin

   write('Натуральное трехзначное число ==>');

   readln(n);

   a := n mod 10;

   b := n div 100;

   c := (n div 100) mod 10;

   l := ( (n mod a) = 0) and ( (n mod b)=0) and ((n mod c)=0);

   writeln(l);

end.
5. По предложенным записям алгоритма напишите программу вычисления значения функции 

[image: image1.png]- ¥ mpx €12,
y=irt o -12<r <0,
x-2mpr 20,




	[image: image2.png]BrmoaY




	Read (x)

If (x–12)

Y:=–x2

else

if (x<0)

Y:=x4

else

Y:= x–2

Write (Y)




Протестируйте полученную программу, используя не менее трех тестовых наборов, чтобы можно было проверить все переходы.

program ch4_5;

var 

   x,y: double;{вещ с двойной точностью}
begin

   write('Значение аргумента ==>');

   readln(x);

   if  x <= -12 then

      y := -x*x

   else

      if x <0 then

         y := x*x*x*x

      else

         y := x-2;

   writeln(y);

end.
6. Даны записи двух алгоритмов подсчета количества нечетных цифр в натуральном числе n. Слева решение оформлено с использованием цикла с предусловием, справа – с постусловием.

Выберите любой способ и переведите его на язык программирования. Протестируйте полученную программу.

Алгоритмы реализуют следующую идею. Из числа выбирается цифра младшего разряда, проверяется на нечетность и затем отбрасывается. Нечётные цифры учитываются. Этот процесс повторяется до тех пор, пока число не исчерпается, т.е. не станет равным нулю.

	Read(n)

K := 0

While ( n <> 0)

if (n mod 10 mod 2 = 1)

K := K + 1

n := n div 10

Write (K)
	Read ( n)

K = 0

Do

if (n mod 10 mod 2 = 1)

K = K + 1

n := n div 10

While (n <> 0 )

Write (K)


program ch4_6_1;

var 

   n: word;

   k: byte;{беззнаковое короткое целое}
begin

   write('Число N ==>');

   readln(n);

   k := 0;

   while  n <> 0 do

   begin

      if  (n mod 10) mod 2 = 1 then

         k := k+1;

      n := n div 10;

   end;

   writeln(k);

end.
program ch4_6_2;

var 

   n: word;

   k: byte;

begin

   write('Число N ==>');

   readln(n);

   k := 0;

   repeat

      if  (n mod 10) mod 2 = 1 then

         k := k+1;

      n := n div 10;

   until n =0;

   writeln(k);

end.
7. Используя предлагаемые записи алгоритма, напишите программу нахождения произведения первых k натуральных чисел, кратных трём.

	Read(k)
P = 1 {здесь накапливаем произведение}

For I := 1 to k/3

P = P * I * 3
Write(P)



 program ch4_7;

var 

   k,i: word;

   p: longword;

begin

   write('Число K ==>');

   readln(k);

   p := 1;

   for i:=1 to k do

      p := p*i*3;

   writeln(p);

end.
8. Напишите программу решения следующей задачи.

Лотерейный номер состоит из трех чисел, каждое из которых может быть от 0 до 99 включительно. Определите шанс угадать выигрышный лотерейный номер.

Идея решения. Надо вычислить количество перестановок для заданного числа объектов (3), которые выбираются из общего числа объектов (100). Перестановка – это любое множество объектов, в котором внутренний порядок имеет значение. Формула определения числа перестановок имеет следующий вид:

Следовательно, надо написать функцию, вычисляющую факториал некоторого числа M (M!=1*2*3*4*…*M) и вычислить количество перестановок.

program ch4_8;

function a_n_k( n,k : word):longword;

var

   i,p : longword;

begin

   p := 1;

   for i := n downto n-k+1 do

      p := p *i;

   a_n_k := p;

end;

begin

   writeln ('Количество сочетаний из 100 по 3: ', a_n_k(100,3));  

end.
9. Напишите программу заполнения массива X(N) положительными и отрицательными элементами. Из него требуется сформировать новый массив только из положительных элементов. Найти из них наибольший элемент. Распечатать новый массив и наибольший элемент.

program ch4_9;

var 

   x,y : array[0..100] of integer;

   i,count : word;

   max : integer;

begin

   for i:=0 to 100 do

      x[i] := random(200)-100;

   count := 0;

   max := -1;

   for i:=0 to 100 do

      if x[i] > 0 then

      begin

         y[count] := x[i];

         if y[count] > max then

            max := y[count];

         inc(count);

      end;

   writeln ('Результат:');

   for i:=0 to count-1 do

      write(y[i],' ');

   writeln;

   writeln ('Максимальный элемент:', max);

end.
10. Напишите программу, в которой предлагается список учащихся разных классов, посещающих футбольную секцию. Требуется упорядочить список по возрастанию классов. Распечатать список фамилий и классов.

program ch4_10;

type

   schooler = record

      fio : string;

      age : word;

   end;

var 

   schoolers : array[1..100] of schooler;

   i,lim : word;

   swap : boolean;

   buf : schooler;

begin

   for i:=1 to 10 do

   begin

      writeln('Ученик №',i);

      write('ФИО ==>');

      readln(schoolers[i].fio);

      write('Класс ==>');

      readln(schoolers[i].age);

   end;

   lim := 9;

   repeat

      swap := false;

      for i:=1 to lim do

      if schoolers[i].age > schoolers[i+1].age then

         begin

            swap := true;

            buf := schoolers[i];

            schoolers[i] := schoolers[i+1];

            schoolers[i+1] := buf;

         end;

         lim := lim-1;

   until not swap;

   writeln ('По возрастанию классов:');

   for i:=0 to 10 do

      write(schoolers[i].fio,', ',schoolers[i].age,' класс.');

end.
11. Напишите программу нахождения минимального элемента первой строки массива К(n, n), заполненного случайными числами.

program ch4_11;

const 

   N = 100;

var 

   k: array[1..N,1..N] of integer;

   i,j,min: byte;

begin

   for i:=1 to N do

      for j:=1 to N do

         k[i,j] := random(10000)-10000;

   min := 1;

   for i:=2 to N do

      if k[1,i] < k[1,min] then

         min := i;

   writeln ('Минимальный элемент первой строки: K[1,',min,']=',k[1,min]);

end.
Эта последняя задача оказалась легче предыдущей, поэтому предлагаем найти минимальный элемент главной диагонали двумерного массива. Усложните таким образом задачу для учащихся, можно не для всех.
Если данных задач недостаточно для повторения основных алгоритмических структур и способов задания массивов, следует воспользоваться сборником задач по информатике для углубленного уровня автора В.Е. Гая, который также входит в УМК углубленного уровня авторов Калинин И.А., Самылкина Н.Н. Это глава 5 "Введение в программирование на языке Паскаль" стр. 267-305. С 305 страницы идут задачи на работу с массивами, которые также можно использовать для отработки темы. Подойдет любой аналогичный задачный материал. Авторы исходят из того, что каждый практикующий учитель имеет свой отработанный задачный материал, который считает наиболее эффективным для изучения темы, поэтому не ставят каких-либо ограничений и условий при обучении программированию.
Темы "Структуры данных" и "Типовые алгоритмы" предлагается изучать следующим образом. В тексте параграфов учебника предлагается теоретический материал и разобраны примеры, иллюстрирующие теорию, но без привязки к конкретной реализации на каком-либо языке программирования. После параграфа предлагаются задания для программирования, которые рекомендуется практически реализовать в изучаемой среде программирования. 
В помощь учителю мы можем предложить решение задач из учебника в виде программ или элементов программного кода.
program task4_18;

Type

  book = record // Структура для хранения данных о конкретной книге на складе

   title : string;

   author: string;

   count : word;

   price : longword;

   place1 : word;

   place2 : word;

   place3 : word;

  end;

  pNode = ^ListNode; // Общая структура для организации учета склада - задача №1 параграфа 18

  ListNode = record

   next : pNode;

   bookData : book;

  end;

// Функция-итератор, подробно рассмотренная в задачнике

  iterate = function( item : pNode): pNode;

// Добавление элемента в список
function itemListNew (title : string; author: string; count : word; price : longword;place1 : word; place2 : word; place3 : word; next : PNode):pNode;

begin

   new (Result);

   Result^.bookData.title := title;

   Result^.bookData.author := author;

   Result^.bookData.count := count;

   Result^.bookData.price := price;

   Result^.bookData.place1 := place1;

   Result^.bookData.place2 := place2;

   Result^.bookData.place3 := place3;

   Result^.next := next;

end;

// Ввод данных о книге с клавиатуры

function itemListKeyboard (next : pNode) : pNode;

var

   title, author: string; 

   count : word; 

   price : longword;

   place1, place2, place3 : word;

begin

   write ('Введите название==>'); readln(title);

   write ('Введите автора==>'); readln(author);

   write ('Введите количество==>'); readln(count);

   write ('Введите цену==>'); readln(price);

   write ('Введите место 1 ==>'); readln(place1);

   write ('Введите место 2==>'); readln(place2);

   write ('Введите место 3==>'); readln(place3);

   Result := itemListNew (title,author,count,price,place1,place2,place3, next);

end;

// Вывод элемента списка на экран

procedure itemListPrint ( item : pNode );

begin


write('|', item^.bookData.title, '|');


write(item^.bookData.author, '|');


write(item^.bookData.count, '|');


write(item^.bookData.price, '|');


write(item^.bookData.place1, '|');


write(item^.bookData.place2, '|');


writeln(item^.bookData.place3, '|');

end;

function itemListDelete(item : pNode) : pNode;

begin


result := item^.next;


Dispose ( item);

end;

procedure iteratorList(start : pNode; iterateFunction : iterate);

var

   current : pNode;

begin

   current := start;

   repeat

      current:=iterateFunction(current);

   until current = nil;

end;

function printListItemAuthor (item : pNode) : pNode;

begin


if item^.bookData.author='Самылкина' then


 writeln(item^.bookData.author,',',item^.bookData.title,',',item^.bookData.price);


result := item^.next;

end;

// Вывод элементов по схеме, оговоренной в задаче

function printListItem (item : pNode) : pNode;

begin


 writeln(item^.bookData.title,',',item^.bookData.price*item^.bookData.count);


 result := item^.next;

end;

var

   head : pNode;

   answ : char;

begin

   // Создание списка
   head := itemListNew ('Информатика - 10','Калинин И.А.,Самылкина Н.Н.',10,210,1,1,1,nil);

   head := itemListNew ('Информатика - 11','Калинин И.А.,Самылкина Н.Н.',10,180,1,1,1,head);

   repeat

      write('Желаете ввести новую книгу?(y/n)');read(answ);

      if answ = 'y' then

         head := itemListKeyboard(head);

   until answ <> 'y';

   iteratorList(head,printListItem); //Вывод всех значений списка - решение задачи 2.

   iteratorList(head, printListItemAuthor); //Вывод книг конкретного автора автора - решение задачи 3.

   iteratorList(head, itemListDelete); //Освобождение памяти

end.
Следующий элемент-описание структуры данных позволяет организовать описание книг как набора различных атрибутов, содержащих строковые или числовые данные. При этом нужно будет заранее оговорить какой атрибут (Например, "Автор") какой номер будет иметь - и после этого книга может описываться как угодно. Количество конкретных упоминаний каждого атрибута не ограничено, но конечно некоторые из них (например, "Цена" или "Количество") будут упоминаться только один раз.
// Задача №4, параграф 18 - универсальная структура хранения данных. До 65535 типов атрибутов.
pAttribute = ^attribute; 
attribute = record;

   nextAttribute : pAttribute;

   attributeType : word; // Тип атрибута: 1- название, 2-автор,3-ключевое слово...

   attributeNumber : longword; // Значение атрибута - число

   attributeString : longword; // Значение атрибута - строка

end;

pBook = ^BookUniversal;

bookUniversal = record

   next : pBook;

   attributeList :pAttribute; 

end;   

Задача № 5, 6 и 7 опираются на структуру данных "Дерево". Решение задачи №5 - структура, разобранная и в учебнике, и в практикуме:

  pBinaryNode = ^BinaryTreeNode; // Структура - ответ на задание №5 параграфа 18

  BinaryTreeNode = record 

   left,right : pNode;

   queryText : string;

  end;

Задачи №6 и 7 используют другую структуру, которая с точки зрения описания отличается только наличием двух сообщений-критериев выбора. Два сообщения-критерия возникают из-за структурных требований карт-определителей - в них выбор ветви предусматривает описание. Например:
 Молодые неразвернувшиеся листья свернуты по оси спирально, в форме «за витка» или «улитки». Листья крупные, рассечены на мелкие доли. По краю нижней стороны листовых долей кучками или рядами расположены спорангии — Папоротники (Polypodiophyta)

• Тип сложения молодых неразвернувшихся листьев иной. Спорангии отсутствуют.

Если узел является листом, то есть приводит к однозначному решению, то в нем должен быть заполнены только один комментарий, левый. При этом дальнейшие переходы - пустые.

Добавление нового узла ветвления (задача №7) в такой ситуации мы рассматриваем как ввод обоих признаков в узел, который ранее был листом и однозначных выводов к этим признакам.
program 4_18_67;

  pNode = ^TreeNode; //Задачи №6 и №7 из параграфа 18

  TreeNode = record 

   left,right : pNode;

   leftText, rightText : string;

  end;

function itemListNew (leftText : string ):pNode;

begin

   new (Result);

   Result^.leftText := leftText;

   Result^.left := nil;

   Result^.right := nil;

end;

procedure newNode ( current : pNode); //Задача №7 из параграфа 18
var

   leftText, rightText : string;

   answ : char;

begin

   write('Ввведите признак первого варианта '); read(leftText);

   write('Ввведите признак второго варианта '); read(rightText);

   current^.leftText := leftText;

   current^.rightText := rightText;

   write('Ввведите вывод по первому варианту '); read(leftText);

   current^.left := itemListNew (leftText);

   write('Ввведите вывод по второму варианту '); read(rightText);

   current^.right := itemListNew (rightText);

end;

var 

   answ : char;

   head,current : pNode;

begin

   current := head;

   repeat

      while current <> nil do

      begin

         if current^.leftText <> '' then

            writeln('Вариант 1: ', current^.leftText);

         if current^.leftText <> '' then

            writeln('Вариант 2: ', current^.rightText);

         if (current^.left <> nil) and (current^.right <> nil) then

         begin

            write('Выберите вариант 1 или 2 '); read(answ);

            if answ = 1 then

               current := current^.left

            else

               current := current^.right;

         end

         else

         begin

            write ('У нас есть только один вариант:',current^.leftText);

            write('Желаете добавить ветвление? (д/н) '); read(answ);

            if answ = 'д' then

               newNode (current);

         end;

      end;

      write('Желаете повторить определение? (д/н) '); read(answ);

    until answ = 'н';

end.
С точки зрения организации работы эту программу нельзя назвать удобной: как минимум, нам было бы необходимо сохранять и читать карты из файла на диске. Кроме того, интерфейс в котором для добавления узла нужно пройти всю процедуру определения нельзя назвать оптимальным (и прямое воспроизведение карты "бумажного" определителя тоже не очень удобно для работы). Тем не менее, это решение задачи - поскольку ничего сверх написанного там не требуется.
В параграфе 19 первой предложена задача на организацию сортировки. При этом есть несколько существенных условий:

1. Данные поступают с клавиатуры, т.е. скорость выполнения операций не критична (все равно львиная доля времени уходит на ввод).

2. Количество данных заранее неизвестно, поэтому организовывать работу с массивом - не выгодно.

Предлагаем вариант решения, опирающийся на адаптированную сортировку вставками. Основной неоптимальный этап этого алгоритма - сдвиг - мы обходим, используя динамический список. Таким образом, алгоритм становится таким же оптимальным, как и сортировка "оптимизированным" пузырьком, причем не требует заранее определить количество сортируемых элементов и "прячет" все операции между вводом новых слов:

program sortWord_0q;

Type

  pNode = ^ListNode; 

  ListNode = record

   next : pNode;

   wrdData : string;

  end;

// Функция-итератор, подробно рассмотренная в задачнике

  iterate = function( item : pNode): pNode;

// Добавление элемента в список
function itemListNew (wordData : string;  next : PNode):pNode;

begin

   new (Result);

   Result^.wrdData := wordData;

   Result^.next := next;

end;

function itemListPrint ( item : pNode ) : pNode;

begin


writeln(item^.wrdData);


result := item^.next;

end;

function itemListDelete(item : pNode) : pNode;

begin


result := item^.next;


Dispose ( item);

end;

procedure iteratorList(start : pNode; iterateFunction : iterate);

var

   current : pNode;

begin

   current := start;

   repeat

      current:=iterateFunction(current);

   until current = nil;

end;

function sortAdd( head : pNode; wrd : string):pNode;

var 

   current, prev : pNode;

begin

   if  head = nil then

         result :=itemListNew(wrd,nil)

   else

   begin

      if (head^.wrdData > wrd) then

         result := itemListNew(wrd,head)

      else

      begin

         current := head;

         prev:=nil;

         while (current <> nil) and (current^.wrdData < wrd) do

         begin

            prev := current;

            current := current^.next;

         end;

         result := head;      

         if current = nil then

            prev^.next := itemListNew(wrd,nil)

         else

            if current^.wrdData <> wrd then

               if prev <> nil then

                  prev^.next := itemListNew(wrd,current)

      end;

   end;

end;

var

   head : pNode;


wrd : string;

 begin

 head := nil;

   write('Введите новое слово:');readln(wrd);

   repeat

      head := sortAdd(head,wrd);

      write('Введите новое слово:');readln(wrd);

   until (length(wrd) < 3)or(length(wrd) > 15);

   iteratorList (head,itemListPrint);  

   iteratorList (head,itemListDelete);

 end.
Простейшее решение - разбить все поступающие слова на файлы, а потом отсортировать их внутри файлов. Количество файлов будет зависеть от того, сколько предполагается слов. Чтобы не затрудняться с сортировкой, можно брать для названий файлов одну, две или три первых буквы поступающих слов. Если поступают буквы, которые нельзя использовать в имени файла - брать их шестнадцатеричные коды.

Тогда сортировка выполняется в два этапа: "раскидать" слова по файлам - используя память только на дескрипторы (причем их даже не обязательно открывать все сразу), а потом считывать слова из файла в память, сортировать (исключая повторы - например, как предложено в задаче 1), и выводить результат.
Решение задачи №3 опирается на предложенный алгоритм хэширования, реализация которого разобрана в практикуме. Отличия будут касаться в первую очередь дополнительных данных, которые мы будем хранить в структуре узла хэш-записи. Там будет храниться список документов, связанных со словом. Поскольку в этой задаче мы не выполняем с этим список никаких операций, то будем хранить его просто в виде строки.
program hashedSearch;

const

   NumHashLen = 577;

type

  pHashNode = ^HashNode;

  HashNode = record

    val : string[40];

    docList : string;

    next : ^HashNode;

  end;

function hashBKDR(S:string) : word; // Хэш-функция
var

   i : byte;

   h  : uint64;

begin

  h := 0;

  for i := 1 to length(s) do

    h := h*61 + ord(s[i]); 

  hashBKDR := h mod NumHashLen;

end;

function addNode( wrd : string; head: pHashNode; list : string ) : pHashNode;

begin

   new(result);

   Result^.val := wrd;

   Result^.docList := list;

   Result^.next := head;

end;

function searchword( wrd : string; head: pHashNode): pHashNode;

begin

   result := head;

   while (result <> nil)and(result^.val <> wrd) do

      result := result^.next;

end;

var

  SearchArray : array[0..NumHashLen] of pHashNode;

  txt : PABCSystem.Text;

  hs  :  word;

  n : pHashNode;

  s,wrd : string;

begin

   for hs:=0 to NumHashLen do // Инициализация пустого массива
      SearchArray[hs]:= nil;

  Assign(txt,'dictionary.txt');

  Reset(txt);

  while not Eof(txt) do

  begin

    ReadLn(txt,s);// Читаем по одному слову

    if length(s) > 1 then

    begin

      wrd := s.Split(':')[0];

      hs := hashBKDR(wrd); // Вычисляем хэш
      SearchArray[hs] := addNode( wrd, SearchArray[hs],s.Split(':')[1] );

    end;  

  end;

  close(txt);

  repeat

   write ('Введите слово для поиска');readln(wrd);

   if length(wrd) > 0 then

   begin

      hs := hashBKDR(wrd); // Вычисляем хэш
      n := searchword(wrd, SearchArray[hs]);

      if n <> nil then

         writeln('Список документов:',n^.docList)

      else

         writeln('Такого слова в списке нет');

   end;

  until wrd = '';

end.

Файл, в котором должен находиться инвертированный список слов, предполагает текстовый формат, в котором каждое слово записано на отдельной строке и отделяется от списка документов двоеточием. Например

математика:учебник1.doc,учебник2.doc,задачник.doc
Четвертая задача будет отличаться от представленной программы тем, что:

1. Разобьет введенную строку на отдельные слова

2. Получит список документов из каждой строки

3. Составит список документов, которые встречаются во всех списках.
program hashedSearchWords;

const

   NumHashLen = 577;

type

  pHashNode = ^HashNode;

  HashNode = record

    val : string;

    docList : string[40];

    next : ^HashNode;

  end;

  pWordNode = ^WordNode;

  WordNode = record

    vl : string;

    next : ^WordNode;

    mark:boolean;

  end;

  iterate = function( item : pWordNode): pWordNode;

function hashBKDR(S:string) : word; // Хэш-функция
var

   i : byte;

   h  : uint64;

begin

  h := 0;

  for i := 1 to length(s) do

    h := h*61 + ord(s[i]); 

  hashBKDR := h mod NumHashLen;

end;

function addNode( wrd : string; head: pHashNode; list : string ) : pHashNode;

begin

   new(result);

   Result^.val := wrd;

   Result^.docList := list;

   Result^.next := head;

end;

function searchword( wrd : string; head: pHashNode): pHashNode;

begin

   result := head;

   while (result <> nil)and(result^.val <> wrd) do

      result := result^.next;

end;

function iterateWordList(start : pWordNode; iter : iterate ):pWordNode;

var

   current : pWordNode;

begin

   current := start;

   while current <> nil do

      current := iter(current);

end;

function doPrint(nod : pWordNode):pWordNode;

begin

   writeln( nod^.vl );

   result := nod^.next;

end;

function doMarcFalse(nod : pWordNode):pWordNode;

begin

   nod^.mark := false;

   result := nod^.next;

end;

function Intersection(head :pWordNode; docs : string ):pWordNode;
//Пересечение списков документов
var

   current,prev : pWordNode;

   wrd:string;

begin

   iterateWordList(head,doMarcFalse);

   foreach wrd in docs.Split(',') do//Помечаем все документы, которые есть в списке
   begin

      current := head;

      while current <> nil do

      begin

         if current^.vl = wrd then

            current^.mark := true;

         current := current^.next;

      end;

   end;

   current := head; prev := nil;

   while current <> nil do

   begin

      if not current^.mark then // Удаляем документы, которых не оказалось в новом списке
      begin

         if prev <> nil then

         begin

            prev^.next:=current^.next;

            Dispose(current);

            current := prev^.next;

         end

         else

         begin

            prev:=current^.next;

            Dispose(current);

            current := prev;

            result := current;

            prev := nil;

         end;

      end

      else

      begin

         prev := current;

         current := current^.next;

      end;

   end;

end;

function SplitList(docs : string) :pWordNode;

var

   wrd : string;

   current : pWordNode;

begin

 foreach wrd in docs.Split(',') do

   begin

      if result = nil then

      begin

         new(current);

         result := current

      end

      else

      begin

         new(current^.next);

         current := current^.next;

      end;

      current^.next := nil;

      current^.vl := wrd;

   end;

end;

var

  SearchArray : array[0..NumHashLen] of pHashNode;

  txt : PABCSystem.Text;

  hs  :  word;

  n : pHashNode;

  s,wrd : string;

  resSearch : pWordNode;

begin

   for hs:=0 to NumHashLen do // Инициализация пустого массива
      SearchArray[hs]:= nil;

  Assign(txt,'dictionary.txt');

  Reset(txt);

  while not Eof(txt) do

  begin

    ReadLn(txt,s);// Читаем по одному слову

    if length(s) > 1 then

    begin

      wrd := s.Split(':')[0];

      hs := hashBKDR(wrd); // Вычисляем хэш
      SearchArray[hs] := addNode( wrd, SearchArray[hs],s.Split(':')[1] );

    end;  

  end;

  close(txt);

  repeat

   write ('Введите слова для поиска:');readln(s);

   if length(s) > 0 then

   begin

      foreach wrd in s.Split(' ') do

      begin

         hs := hashBKDR(wrd); // Вычисляем хэш
         n := searchword(wrd, SearchArray[hs]);

         if n <> nil then

         begin

            if resSearch <> nil then

               resSearch := Intersection(resSearch,n^.docList)

            else

               resSearch := SplitList(n^.docList);

         end

         else

            break;

      end;

      if resSearch <> nil then

         iterateWordList(resSearch,doPrint)

      else

         WriteLn('Извините, таких документов в списке нет.');

   end;

  until wrd = '';

end.

